
Apéndice 1
Sintaxis de Modula-2

En este apéndice se recogen de forma precisa todas las reglas sintácticas que definen el lenguaje
Modula-2, utilizando la notación BNF (Backus-Naur Form). En las reglas se utilizan los siguientes
metasímbolos.

Metasímbolo Uso
Metasímbolo de definición. Indica que el elemento a su izquierda puede
desarrollarse según el esquema de la derecha.
Metasímbolo de alternativa. Indica que puede elegirse uno y sólo uno de los
elementos separados por este metasímbolo.
Metasímbolo de repetición. Indican que los elementos incluidos dentro de ellos
se pueden repetir cero o más veces.
Metasímbolo de opción. Indican que los elementos incluidos dentro de ellos
pueden ser utilizados o no.
Metasímbolo de agrupación. Agrupan los elementos incluidos en su interior.

Tabla A1-01: Definición y uso de los Metasímbolos.

Estos metasímbolos se escriben en el tipo de letra especial indicado para distinguirlos de los paréntesis,
corchetes, etc. que forman parte del lenguaje Modula-2. También se emplearán distintos estilos de
letra para distinguir los elementos simbólicos siguientes:

Estilo Uso

Elemento_no_terminal
Este estilo se emplea para escribir el nombre de un elemento gramatical
que habrá de ser definido por alguna regla. Cualquier elemento a la
izquierda del metasímbolo  será no terminal y aparecerá con este estilo.

Elemento_terminal
Este estilo se emplea para representar los elementos que forman parte del
lenguaje Modula-2, es decir, que constituyen el texto de un programa. Si
aparecen en una regla deberán escribirse exactamente como se indican.

Tabla A1-02: Estilos de letras y uso.

La mayoría de las reglas ya han sido introducidas a lo largo del texto. Sin embargo, algunas de ellas se
refieren a ciertos aspectos del lenguaje que se salen fuera de los objetivos de un primer curso de
programación. Pese a todo y con la idea de ofrecer una guía de consulta rápida del lenguaje, se ha
preferido realizar una definición conjunta de todas las reglas del lenguaje desde la regla más global
hasta la más particular. Estas reglas, agrupadas por los grandes elementos del lenguaje, son las
siguientes:

Nº Reglas de la Unidad de Compilación
1 Unidad_de_compilación ::=

   Módulo_programa |
   Módulo_definición |
   Módulo_implementación

2 Módulo_programa ::=
   Cabecera_módulo
   Bloque
   Identificador .



Apéndice 1: Sintaxis de Modula-2

Página 2 de 8

3 Módulo_definición ::=
   Cabecera_definición
   { Definición_de_módulo }
   END Identificador .

4 Módulo_implementación ::=
   IMPLEMENTATION Módulo_programa

5 Cabecera_módulo ::=
   MODULE Identificador
   [ Prioridad ] ;
   { Lista_importados ; }

6 Cabecera_definición ::=
   DEFINITION MODULE Identífícador;
   { Lista_importados ; }
   [ Lista_exportados ; ]

7 Lista_importados ::=
   [ FROM Identificador ]
   IMPORT Lista_de_identificadores

8 Lista_exportados ::=
   EXPORT [ QUALIFIED ] Lista_de_identificadores

9 Prioridad ::=
   [ Expresión_constante ]

10 Bloque ::=
   Parte_declarativa
   Parte_ejecutiva
   END

11 Definición_de_módulo ::=
   Declaración_de_constantes |
   TYPE { Identificador [ = Esquema_de_tipo ] ; } |
   Declaración_de_variables |
   Cabecera_subprograma ;

Nº Reglas de Declaraciones
12 Parte_declarativa ::=

   { Declaración }

13 Declaración ::=
   Declaración_de_constantes | Declaración_de_tipos |
   Declaración_de_variables | Declaración_subprograma |
   Declaración_módulo

14 Declaración_de_constantes ::=
   CONST { Asociación_constante ; }

15 Asociación_constante :: =
   Identificador = Expresión_constante

16 Declaración_de_tipos ::=
   TYPE { Definición_de_tipo ; }

17 Definición_de_tipo ::=
   Identificador = Esquema_de_tipo

18 Esquema_de_tipo ::=
   Tipo_simple | Tipo_conjunto | Tipo_formación |
   Tipo_registro | Tipo_puntero | Tipo_subprograma

19 Tipo_simple ::=
   Identificador_de_tipo | Tipo_enumerado | Tipo_subrango

20 Tipo_enumerado ::=
   ( Lista_de_identificadores )

21 Tipo_subrango ::=
   [ Identificador_de_tipo ]
   [ Expresión_constante .. Expresión_constante ]

22 Tipo_conjunto ::=
   SET OF Tipo_simple

23 Tipo_formación ::=
   ARRAY Tipo_simple { , Tipo_simple } OF Esquema_de_tipo

24 Tipo_registro ::=
   RECORD Secuencia_de_listas_de_campos END

25 Secuencia_de_listas_de_campos ::=
   Lista_de_campos { ; Lista_de_campos }

26 Lista_de_campos ::=
   [ Lista_de_identificadores : Esquema_de_tipo |
     Parte_variante ]

27 Parte_variante ::=
   CASE [ Identificador ] : Identificador_de_tipo OF
   Variante { | Variante }
   [ ELSE Secuencia_de_listas_de_campos ]
   END

28 Variante ::=
   Lista_de_valores : Secuencia_de_listas_de_campos

29 Tipo_puntero ::=
   POINTER TO Esquema_de_tipo

30 Tipo_subprograma ::=
   PROCEDURE [ Lista_de_tipos_formales ]

31 Lista_de_tipos_formales ::=
   ( [ [ VAR ] Tipo_formal { ,[ VAR ] Tipo_formal } ] )
   [ : Identificador_de_tipo ]

32 Declaración_de_variables ::=
   VAR { Lista_de_variables ; }

33 Lista_de_variables ::=
   Lista_de_identificadores : Esquema_de_tipo

34 Declaración_subprograma ::=
   Cabecera_subprograma ;
   Bloque
   Identificador ;

35 Cabecera_subprograma ::=
   PROCEDURE Identificador [ Parámetros_formales ]

36 Parámetros_formales ::=
   ( [ Grupo_de_parámetros { ; Grupo_de_parámetros } ] )
   [ : Identificador_de_tipo ]

37 Grupo_de_parámetros ::=
   [ VAR ] Lista_de_Identificadores : Tipo_formal

38 Declaración_módulo ::=
   MODULE Identificador
   [ Prioridad ]
   { Lista_importados ; }
   [ Lista_exportados ; ]
   Bloque
   Identificador ;

Nº Reglas de Sentencias Ejecutables
39 Parte_ejecutiva ::=

  [ BEGIN Secuencia_de_sentencias ]

40 Secuencia_de_sentencias ::=
   Sentencia { ; Sentencia }

41 Sentencia ::=
   [ Sentencia_de_asignación |
     Sentencia_de_llamada_a_procedimiento | Sentencia_IF |
     Sentencia_CASE | Sentencia_WHILE |
     Sentencia_REPEAT | Sentencia_LOOP |
     Sentencia_FOR | Sentencia_WITH |
     Sentencia_RETURN | Sentencia_EXIT ]

42 Sentencia_de_asignación ::=
   Variable := Expresión

43 Sentencia_de_llamada_a_procedimiento ::=
   ( Identificador_de_procedimiento | Variable )
   [ Parámetros_de_llamada ]

44 Sentencia_IF ::=
   IF Expresión THEN Secuencia_de_sentencias
   { ELSIF Expresión THEN Secuencia_de_sentencias }
   [ ELSE Secuencia_de_sentencias ]
   END

45 Sentencia_CASE ::=
   CASE Expresión OF
   Caso { | Caso }
   [ ELSE Secuencia_de_sentencias ]
   END

46 Caso ::=
   Lista_de_valores : Secuencia_de_sentencias

47 Sentencia_WHILE ::=
   WHILE Expresión DO Secuencia_de_sentencias END

48 Sentencia_REPEAT ::=
   REPEAT Secuencia_de_sentencias UNTIL Expresión

49 Sentencia_LOOP ::=
   LOOP Secuencia_de_sentencias END

50 Sentencia_FOR ::=
   FOR Identificador_de_variable := Expresión TO Expresión
   [ BY Expresión_constante ] DO Secuencia_de_sentencias
   END

51 Sentencia_WITH ::=
   WITH Variable DO Secuencia_de_sentencias END

52 Sentencia_RETURN ::=
   RETURN [ Expresión ]

53 Sentencia_EXIT ::=
   EXIT

Nº Reglas de Expresiones
54 Lista_de_valores ::=

   Valores { , Valores }

55 Valores ::=
   Expresión_constante [ .. Expresión_constante ]

56 Expresión_constante ::=
   Expresión_constante_simple
   [ Operador_Comparador Expresión_constante_simple ]

57 Expresión_constante_simple ::=
   [ + | - ] Término_constante
   { Operador_sumador Término_constante }

58 Término_constante ::=
   Factor_constante
   { Operador_multiplicador Factor_constante }

59 Factor_constante ::=
   Identificador_de_constante | Número | Ristra | Conjunto |
   ( Expresión_constante ) | NOT Factor_constante |
   Carácter_en_octal

60 Parámetros_de_llamada ::=
   ( [ Lista_de_expresiones ] )

61 Lista_de_expresiones ::=
   Expresión { , Expresión }

62 Expresión ::=
   Expresión_simple
   [ Operador_comparador Expresión_simple ]

63 Expresión_simple ::=
   [ + | - ] Término { Operador_sumador Término }

64 Término ::=
   Factor { Operador_multiplicador Factor }

65 Factor ::=
   Variable | Identificador_de_procedimiento |
   Identificador_de_función | Identificador_de constante |
   Número | Ristra | Conjunto | ( Expresión ) |
   NOT Factor | Carácter_en_octal | Llamada_a_función

66 Llamada_a_función ::=
   ( Identificador_de_función | Variable )
   [ Parámetros_de_llamada ]

67 Operador_comparador ::=
   = | <> | # | < | <= | > | >= | IN

68 Operador_sumador ::=
   + | - | OR

69 Operador_multiplicador ::=
   * | / | DIV | MOD | & | AND

Nº Reglas de Elementos Básicos
70 Tipo_formal ::=

   [ ARRAY OF ] Identificador_de_tipo

71 Variable ::=
   ( Identificador_de_variable | Identificador_de_campo )
   { . Identificador_de_campo | [ Lista_de_expresiones ] | ^ }

72 Conjunto ::=
   [ Identificador_de_tipo ] { Lista_de_elementos }

73 Lista_de_elementos ::=
   [ Elementos { , Elementos } ]

74 Elementos ::=
   Expresión_constante [ .. Expresión_constante ]

75 Ristra ::=
   " { Carácter } " | ' { Carácter } '

76 Número ::=
   Número_entero | Número_real

77 Número_entero ::=
   dígito { dígito } | dígito_octal { dígito_octal } B |
   dígito { dígito_hexadecimal } H

78 dígito_hexadecimal ::=
   dígito | A | B | C | D | E | F

79 dígito ::=
   dígito_octal | 8 | 9

80 dígito_octal ::=
   0 | 1 | 2 | 3 | 4 | 5 | 6 | 7

81 Número_real ::=
   dígito { dígito } . { dígito } [ escala ]

82 escala ::=
   E [ + | - ] dígito { digito }

83 Carácter_en_octal ::=
   dígito_octal { dígito_octal } C

84 Lista_de_identificadores ::=
   Identificador { , Identificador }

85 Identificador_de_XXXX ::=
   Identificador { . Identificador }

XXXX puede ser igual a constante, variable, tipo, campo, procedimiento o función
86 Identificador ::=

   letra { letra | dígito }

87 letra ::=
   A | B | C | D | E | F | G | H | I | J | K | L | M |
   N | O | P | Q | R | S | T | U | V | W | X | Y | Z |
   a | b | c | d | e | f | g | h | i | j | k | l | m |
   n | o | p | q | r | s | t | u | v | w | x | y | z

Tabla A1-03: Reglas

Indice de elementos sintácticos por orden alfabético

Reglas Nº
Asociación_constante 15
Bloque 10
Cabecera_definición 6
Cabecera_módulo 5
Cabecera_subprograma 35
Carácter_en_octal 83
Caso 46
Conjunto 72
Declaración 13
Declaración_de_constantes 14
Declaración_de_tipos 16
Declaracíón_de_variables 32
Declaración_módulo 38
Declaración_subprograma 34
Definición_de_módulo 11



Programación I

Página 7 de 8

Definición_de_tipo 17
dígito 79
dígito_hexadecimal 78
dígito_octal 80
Elementos 74
escala 82
Esquema_de_tipo 18
Expresión 62
Expresión_constante 56
Expresión_constante_simple 57
Expresión_simple 63
Factor 65
Factor_constante 59
Grupo_de_parámetros 37
Identificador 86
Identificador_de_campo 85
Identificador_de_constante 85
Identificador_de_función 85
Identificador_de_procedimiento 85
Identificador_de_tipo 85
Identificador_de_variable 85
letra 87
Lista_de_campos 26
Lista_de_elementos 73
Lista_de_expresiones 61
Lista_de_identificadores 84
Lista_de_tipos_formales 31
Lista_de_valores 54
Lista_de_variables 33
Lista_exportados 8
Lista_importados 7
Llamada_a_función 66
Módulo_definición 3
Módulo_implementación 4
Módulo_programa 2
Número 76
Número_entero 77
Número_real 81
Operador_comparador 67
Operador_multiplicador 69
Operador_sumador 68
Parámetros_de_llamada 60
Parámetros_formales 36
Parte_declarativa 12
Parte_ejecutiva 39
Parte_variante 27
Prioridad 9

Secuencia_de_listas_de_campos 25
Secuencia_de_sentencias 40
Sentencia 41
Sentencia_CASE 45
Sentencia_de_asignación 42
Sentencia_de_llamada_a_procedimiento 43
Sentencia_EXIT 53
Sentencia_FOR 50
Sentencia_IF 44
Sentencia_LOOP 49
Sentencia_REPEAT 48
Sentencia_RETURN 52
Sentencia_WHILE 47
Sentencia_WITH 51
Término 64
Término_constante 58
Tipo_conjunto 22
Tipo_enumerado 20
Tipo_formación 23
Tipo_formal 70
Tipo_puntero 29
Tipo_registro 24
Tipo_simple 19
Tipo_subprograma 30
Tipo_subrango 21
Ristra 75
Unidad_de_compilación 1
Valores 55
Variable 71
Variante 28
Tabla A1-04: Indice de elementos sintácticos por orden alfabético


